# $Id: PKGBUILD 83062 2013-01-27 09:45:44Z andrea $
# Maintainer: Kevin Piche <kevin@archlinux.org>
# Contributor: Jochem Kossen <j.kossen@home.nl>

pkgname=evilwm
pkgver=1.1.0
pkgrel=1
pkgdesc="A minimalist but usable window manager for the X Window System."
arch=('i686' 'x86_64')
url="http://evilwm.sourceforge.net"
license=('custom')
depends=('libxrandr')
source=(http://www.6809.org.uk/evilwm/$pkgname-$pkgver.tar.gz LICENSE)
md5sums=('5f14cc3019320860c85bb07c60a76bd7'
         '0941ad3e96597ef75c6310f43a0342a0')

build() {
  cd "$srcdir"/$pkgname-$pkgver
  make
}

package() {
  cd "$srcdir"/$pkgname-$pkgver
  make DESTDIR="$pkgdir" desktopfilesdir=/usr/share/xsessions install
  install -D -m644 "$srcdir"/LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
}
